

### Gravitációs gyorsulás mérése

Végezetül bemutatjuk, hogy egy egyszerű, középiskolákban is jól ismert mérési problémát hogyan lehet számítógépre „ültetni”.

Feladatul tűzzük ki azt, hogy az iskolaszámítógéppel vezérelt ejtési kísérletek (szabadesés) alapján megmérjük a gravitációs gyorsulást.

Mint ismeretes, a zérus kezdősebességű, h magasságból eső test t esési ideje és a g gravitációs gyorsulás közti kapcsolat (légüres térben):

$$g = \frac{2 \cdot h}{t^2} \quad (1)$$

Nézzük meg a pontossági lehetőségeinket. (1) teljes differenciáljából g relatív hibájának abszolút értéke:


$$\left| \frac{\Delta g}{g} \right| = \left| \frac{\Delta h}{h} \right| + 2 \cdot \left| \frac{\Delta t}{t} \right| \quad (2)$$

#### 8. PROGRAM

```
10 CLS
20 PRINT TAB(20) "SZABADESES - G MERESE"
30 OUT 31,7: OUT 30,64: OUT 31,14: OUT 30,255
40 PRINT "TEGYE FEL AZ ELEKTROMAGNES FEGYVERZETERE"
50 PRINT " A CSAPAGYOLYOT!": PRINT: I=0
60 INPUT "A SZABADESES MAGASSAGA M-BEN=":H
70 OUT 30,0: OUT 31,15
80 I=I+1: IF INP(31)=255 THEN 80
90 T=(I+4.5)/92.3: N=N+1: G=2*H/T/T
100 PRINT "N=";N,"H=";H,"T=";T,"G=";G: GOTO 30
```

#### 9. PROGRAM

```
10 CLS: DEFINT I,M,N
20 PRINT TAB(20) "SZABADESES - G MERESE"
30 PRINT TAB(20) "XXXXXXXXXXXXXXXXXXXXX"
40 POKE 16526,0: POKE 16527,125
50 FOR I=32000 TO 32027: READ M: POKE I,M: NEXT I
60 PRINT: INPUT "A GOLYO ATMEROJE MM-BEN=":D1: D=D1/1000
70 N=N+1
80 OUT 31,7: OUT 30,64: OUT 31,14: OUT 30,255
90 PRINT: PRINT "TEGYE FEL A CSAPAGYOLYOT"
100 PRINT "AZ ELEKTROMAGNES FEGYVERZETERE!"
110 PRINT: INPUT "A SZABADESES MAGASSAGA M-BEN=":H
120 I=USR(0): IF I<0 THEN 130 ELSE 140
130 PRINT: PRINT "A GOLYO NEM AZ ERZEKELORE ESETT!": GOTO 80
140 T=(55*I+18)/1750000: G=2*H/T/T
150 GK=G*(1+1.25/78000*(1+0.225/2*H/D)): G0=G0+GK: A=G0/N
160 PRINT "N=";N,"H=";H,"M=";M,"T=";T,"S"
170 PRINT "G=";G,"GK=";GK,"A=";A: GOTO 70
180 REM LD HL,0 - LD A,0 - OUT (1E),A - LD A,0F - OUT (1F),A
190 DATA 33,0,0,62,0,211,30,62,15,211,31
200 REM INC HL - LD A,H - ADD A,00 - JP Z,7D19 - IN A, (1F)
210 DATA 35,124,198,128,202,25,125,219,31
220 REM ADD A,1 - JP Z,7D0B - JP 0A9A
230 DATA 198,1,202,11,125,195,154,10
```



6. ábra

Tanteremben végzett kísérleteknél a szóba jöhető ejtési magasság 1-2

méter, amit pl. acél mérőszalaggal 1 mm pontossággal tudunk mérni, azaz

$$\left| \frac{\Delta h}{h} \right| = 10^{-3} \quad (3)$$

Az időmérésből adódó relatív hibát sem célszerű sokkal  $10^{-3}$  alá felvenni, ami t relatív hibájára a

$$\left| \frac{\Delta t}{t} \right| = 5 \cdot 10^{-4} \quad (4)$$

korlátot adja.

h = 1 m esetén az esési idő kb. fél másodperc, ami (4) alapján azt jelenti, hogy ezt 250  $\mu$ s pontossággal kell mérnünk. Kézi működtetésű stopperrel természetesen ez nem

#### 10. PROGRAM

```
10 CLS: DEFINT R: FO=1.78E+6
20 PRINT TAB(16) "AZ ORAJEL FREKVENCIAJANAK MERESE"
30 PRINT TAB(20) "HANGVILLA SEGITSEGEVEL"
40 OUT 31,7: OUT 30,254: OUT 31,8: OUT 30,15
50 PRINT: INPUT "A HANGVILLA FREKVENCIAJA HZ-BEN=":F
60 PRINT: PRINT "NYOMJA LE A <.> VAGY A <,> BILLENTYUT!"
70 D=FO/16/F+0.5: R1=D/256: R0=D-256*R1
80 I0=INKEY$
90 IF I0="" THEN R0=R0+1 ELSE IF I0="," THEN R0=R0-1 ELSE 80
100 IF R0<0 THEN 110 ELSE IF R0>255 THEN 120 ELSE 130
110 R0=255: R1=R1-1: GOTO 130
120 R0=0: R1=R1+1
130 OUT 31,0: OUT 30,R0: OUT 31,1: OUT 30,R1
140 FO=16*(R0+256*R1)*F
150 PRINT@ 8*64,"F-ORA=";FO: GOTO 80
```

#### 11. PROGRAM

```
10 CLS: DEFINT A,H,I,M
20 PRINT TAB(10) "AZ ORAJEL FREKVENCIAJANAK MERESE STOPPERREL"
30 FOR H=1 TO 4: READ I,A: OUT 31,I: OUT 30,A: NEXT H
40 POKE 16526,0: POKE 16527,125
50 FOR I=32000 TO 32031: READ M: POKE I,M: NEXT I
60 PRINT: PRINT "A STOPPER INDITASAVAL EGYUTT"
70 PRINT "NYOMJA LE VALAMELYIK BILLENTYUT!"
80 PRINT: PRINT "KB. 18 PERC ELTELTVE A HANGJELZESRE"
90 PRINT "ALLITS A STOPPERT!"
100 I0=INKEY$: IF I0="" THEN 100 ELSE 110
110 X=USR(0): FOR I=0 TO 100: NEXT I: OUT 30,0
120 PRINT: INPUT "A MERT IDO MASODPERCEN=":TM
130 F=1.88642E+9/TM: T=1/F: RESTORE
140 PRINT "F=";F,"HZ","T=";T,"S": GOTO 30
150 DATA 0,248,1,0,7,254,8,15: REM HANGGENERATOR VEZERLESE
160 REM LD BC,0 - LD H,0 - LD L,F8 - LD A,0 - OUT (1E),A
170 DATA 1,0,0,38,0,46,248,62,0,211,30
180 REM INC B - JP NZ,7D0B - INC C - JP NZ,7D0B
190 DATA 4,194,11,125,12,194,11,125
200 REM INC H - JP NZ,7D0B - INC L - JP NZ,7D0B
210 DATA 36,194,11,125,44,194,11,125
220 REM LD A,0F - OUT (1E),A - RET
230 DATA 62,15,211,30,201
```

megy, de számítógépünkkel a probléma könnyen megoldható.

Az ettelt tárgy legyen egy 8–10 mm átmérőjű acél csapágygolyó, amit a kimenetként működő A csatorna valamelyik bitje által vezérelt elektromágnes fegyverzete tart (6. ábra). Ha a szóban forgó bit értéke 1, az elektromágnes gerjesztődik, tartja a golyót. Ha a bit 0-ra vált, a  $T_1$  és  $T_2$  tranzisztor lezár, az elektromágnes „elengedi” a golyót, az ráesik az alatta levő K érintkezőre, zárja azt, és a bemenetként működő B csatornán keresztül leállítja azt az órát, amelyik a golyó esésekor indult el.

Először készítsük el a 6. ábrán látható áramkört. A tekercs: 8–10 mm átmérőjű, 2–3 cm hosszú ferritdarabra 0,2–0,3 mm átmérőjű zománcszigetelésű rézhuzalból csévéljünk fel kb. 150 menetet.

A K érintkező 0,3–0,4 mm vastag, 2–3 mm széles, 6–8 cm hosszú acéllemezéből készülhet (óra rugó). A  $T_2$  tranzisztor (BSX32 vagy bármilyen, kb. 1 A kollektoráramú típus megfelel) tanácsos hűtőszálalóval ellátni.

Az elektromágneset fogassuk be Bunsen-állványba. A K érintkezőt helyezzük a mágnes alá, attól kb. 1–2 méterre. Írjuk be a 8. PROGRAM-ot, csatlakoztassuk áramkörünk bemenetét az A csatorna, a K érintkezőt pedig a B csatorna valamelyik pontjára. Indítsuk el a programot, majd helyezzük föl a csapágygolyót az elektromágnes fegyverzetére és a 4,7 k $\Omega$ -os trimmerrel úgy állítsuk be  $T_2$  kollektoráramát, hogy a mágnes éppen megtartsa a golyót. Mérjük meg az esési magasságot (a K érintkező zárt helyzetű pozíciója és a golyó alsó része közti távolság), írjuk be ezt, és nyomjuk meg a NEW LINE gombot. Ha mindent jól csináltunk, a golyó esni kezd, elindul az óra (lásd a 80-as sort), majd a K érintkezőre eső golyó leállítja az órát.

A program kiírja a mérés sorszámát, az esési magasságot és időt, valamint az ezekből számított  $g$  értéket. Ezek a gyorsulás értékek azonban meglehetősen nagy szórást mutatnak. Ennek oka az időmérés pontatlansága. BASIC ciklusra épülő óránk felbontóképessége kb. 0,01 s (lásd 90-es sor), ami 1 méteres esési magasságnál majdnem 0,5 m·s<sup>-2</sup>-os bizonytalanságot okoz. Az is előfordulhat — ha a K érintkező nem megfelelő pillanatban van 0,01 s-nál rövidebb ideig zárva —, hogy a 31-es kapuról beolvasott érték mindig 255, azaz az óra nem áll le. Más megoldás után kell néznünk.

Sokkal pontosabb, finomabb felbontású stoppert „készíthetünk”, ha az időmérést a számítógép kvarcvezérlésű órajelének periódusidejére alapozzuk.

A 9. PROGRAM lényegében azonos a 8.-kal, de itt az időmérés a 120-as sorban meghívott gépi nyelvű szubrutinra épül. Ez a szubrutin a kezdőértékek beállítása után

(a HL regiszterpár és az A kapu bitjeinek nullázása) az órajel periódusidejének 55-szöröse alatt (kb. 31  $\mu$ s) egyesével növeli a HL regiszterpár értékét mindaddig, amíg a bemenetként üzemelő B kapuról beolvasott érték 255, azaz amíg a golyó le nem esik a K érintkezőre. Ekkor a rendszer értékátadással visszatér BASIC-be. A 120-as sorban I értéke az a szám lesz, ameddig a golyó esése közben a szubrutin elszámolt. A 140-es sorban I és az órajel frekvenciája alapján (esetünkben 1750000 Hz, de ezt részletesen lásd később) kiszámítjuk az esési időt, majd  $g$  értékét.

A kísérletet minden bizonnyal nem légüres térben végezzük, tehát figyelembe kell vennünk a golyóra ható felhajtóerőt és a közegellenállást is.

Ha  $k$  közegellenállási tényezőjű,  $d$  átmérőjű,  $\rho_k$  sűrűségű golyót ejtünk nem túl nagy  $h$  magasságból  $\rho_1$  sűrűségű közegben, akkor a  $t$  esési idő, a  $g_k$  gravitációs gyorsulás és a fenti paraméterek közötti kapcsolat:

$$g_k \approx \frac{2 \cdot h}{t^2} \left[ 1 - \frac{\rho_1}{\rho_k} \left( 1 - \frac{k \cdot h}{2 \cdot d} \right) \right]. \quad (5)$$

A 150-es sorban GK a gyorsulás fentiek szerint korrigált értéke. A program kiszámítja a mérési eredmények átlagát is (A), majd kiírja a következőket:

- mérés sorszáma (N)
- esési magasság (H)
- esési idő (T)
- $g$  korrigálatlan értéke (G)
- $g$  korrigált értéke (GK)
- korrigált értékek átlaga (A).

Ehhez a programhoz is a 6. ábrán bemutatott összeállítás használható. Ha a golyó az esés kezdetétől számított 1 másodpercen belül nem esik az érzékelőre, a gép ezt jelzi és ez nem okoz mérési hibát. Változatlan ejtési magasság esetén  $h$  értékét nem szükséges mindig újra beírni, elegendő a NEW LINE billentyű megnyomása.

Mint látható, a 9. PROGRAM-mal végzett mérés a számítógép órajelének frekvenciájára épül. Ha biztosítani akarjuk a (4) alatti pontosság követelményt, akkor ezt a frekvenciát is  $5 \cdot 10^{-4}$ -nél kisebb relatív hibával kell ismernünk. A számítógép kezelési könyvében megadott frekvencia csak tájékoztató jellegű.

Ha rendelkezésünkre áll hiteles frekvenciamérő, akkor kössük azt a bővítő csatlakozó 30-as pontjára, mérjük meg a tényleges frekvenciát és a 9. PROGRAM 140-es sorába, a jelenlegi 1750000 helyett, a mért értéket írjuk be.

A középiskolák zömében azonban nincs ilyen műszer, de a 10. vagy a 11. PROGRAM-mal és egy hiteles hangvilla, illetve egy pontos kvarckaróra segítségével a szükséges pontossággal megmérhető az órajel tényleges frekvenciája.

A 10. PROGRAM működésének

alapja a következő. Az órajel frekvenciája ( $f_0$ ), a hanggenerátor által megszólaltatott hang frekvenciája ( $f$ ) és az  $R_0$ ,  $R_1$  regiszterek tartalma közti kapcsolat:

$$f_0 = 16 \cdot (R_0 - 256 \cdot R_1) \cdot f. \quad (6)$$

A számítógéppel ismert  $f$  frekvenciájú hangot előállítva (6) alapján  $f_0$  számítható (ez történik a 140-es sorban).

A program beírása és elindítása után írjuk be a rendelkezésünkre álló hangvilla frekvenciáját (rendszerint 440 Hz), szólaltassuk meg a hangvillát és a „.”, illetve „,” billentyű lenyomásával addig növeljük, illetve csökkentjük a hangszóróból hallható hang frekvenciáját, amíg a hangvilla és a számítógép hangja a lehető legkisebb lebegési frekvenciával szólal meg. Az ekkor kiírt „F-ORA” érték az órajel frekvenciája kb. 3 értékes jegyre.

A 11. PROGRAM-mal sokkal pontosabb mérést végezhetünk. Ebben egy olyan gépi kódú szubrutint hívunk meg, amely az órajel periódusidejének 1,886 42·10<sup>9</sup>-szere se alatt fut le. Ehhez kb. 18 percre van szükség, amit ha stopperrel, kvarccórával 0,5 s bizonytalansággal mérünk, a relatív hiba csak 4,6·10<sup>-4</sup>, azaz az órajel frekvenciáját a kívánt pontossággal megkapjuk.

## További lehetőségek

A két, 8 bites digitális be, kimenetű egységün elcsatlakoztatva 8 bites analóg/digitál konvertereket (pl. HAD 08). Egy ily módon kibővített iskolaszámítógép a középiskolai fizikaszertár modern központi műszerévé válhat.

Néhány alkalmazási példa:

### HANGGENERÁTOR, VOBULÁTOR

27 Hz-től kb. 100 kHz-ig 4096 lépésben 0,5 V-os négyzögjeleket adhatunk ki programozható módon változtatva a frekvenciát;

### JELGENERÁTOR, VOBULÁTOR

0-tól kb. 30 kHz-ig TTL szintű, tetszőleges kitöltési tényezőjű négyzögjelek kiadása a két kapu bármelyik lábára (a frekvencia és a kitöltési tényező is programozhatóan változtatható!);

### OSZCILLOSKÓP, MEMÓRIASZKÓP

a két kapura kötött A/D konverteren keresztül analóg jelek mérése, megjelenítése, feldolgozása.

## Irodalomjegyzék

- BUDÓ Á.: Mechanika, TK., Bp., 1965.  
 IZSAK M.: Híradástechnikai kislexikon, MK., Bp., 1976.  
 SPRACKLEN K.: Z-80 and 8080 Assembly Language Programming, Hayden Book Co., Inc. New Jersey, 1979.  
 ZAKS R.: Programming the Z-80, Sybex Inc. Berkeley, USA, 1980.